
• For Project 3, the SHA256 intermediate values are 
provided in 
 
simplified_sha256.xlsx 
 

• The wt values at each time t are provided in  
 
simplified_sha256_w_values.xlsx 
 

• See links on course schedule page 

Project 3 

Using $display statements 
$display("<format>", exp1, exp2, ...); // formatted write to display 
  format indication %b %B binary  

                    %c %C character (low 8 bits)  

                    %d %D decimal %0d for minimum width field  
                    %e %E E format floating point %15.7E  

                    %f %F F format floating point %9.7F  

                    %g %G G general format floating point  

                    %h %H hexadecimal  
                    %l %L library binding information  

                    %m %M hierarchical name, no expression  

                    %o %O octal  

                    %s %S string, 8 bits per character 
                    %t %T simulation time, expression is $time  

                    %u %U unformatted two value data 0 and 1  

                    %v %V net signal strength  

                    %z %Z unformatted four value data 0, 1, x, z  

// $display will automatically insert a newline 

 

$write // same as $display except no automatic insertion of newline  
$monitor // same as $display except only displays if expression changes 

Using $display statements 
escape sequences, quoted characters in strings 

    \n newline  

    \t tab  

    \\ backslash  

    \" quote  

    \ddd character by octal code of 1-3 digits (0 ≤ d ≤ 7) 
    %% percent 

 

any other characters between the quotes are displayed 

the expressions are taken in order of the format indication 

,, in the expression list inserts one space in the output 

Using $display statements 
function logic [255:0] sha256_op(input logic [31:0] a, b, c, d, e, f, g, h, w, k); 
    logic [31:0] S1, S0, ch, maj, t1, t2; // internal signals 
begin 
    S1 = rightrotate(e, 6) ^ rightrotate(e, 11) ^ rightrotate(e, 25); 
    ch = (e & f) ^ ((~e) & g); 
    t1 = ch + S1 + h + k + w; 
    S0 = rightrotate(a, 2) ^ rightrotate(a, 13) ^ rightrotate(a, 22); 
    maj = (a & b) ^ (a & c) ^ (b & c); 
    t2 = maj + S0; 

    // this displays the next values of a, b, c, d, e, f, g, h 
    // just like the spreadsheet 
    $display("%h %h %h %h %h %h %h %h", t1 + t2, a, b, c, d + t1, e, f, g); 
    sha256_op = {t1 + t2, a, b, c, d + t1, e, f, g}; 
end 
endfunction 

 

always_ff @(...) begin 
    if (!reset_n) begin 
        ... 
    end else begin 
        ... 

        $write("%2d ", t); 
        {a, b, c, d, e, f, g, h} <= sha256_op(a, b, c, d, e, f, g, h, w[15], t); 
        ... 
    end 
end 

Modelsim in Command Line Mode 
To run Modelsim in command line mode, create a work directory:  
 
vlib work  
 
You just have to run this command once to create the work directory.  
 
To compile a design with testbench, run the following (using for 
example tb_fibonacci_calculator.sv and fibonacci_calculator.sv):  
 
vlog tb_fibonacci_calculator.sv fibonacci_calculator.sv  
 
If there are other SystemVerilog files, include them on this command line a well.  
 
To simulate, run the following (again using the fibonacci calculator example):  
 
vsim -c -do "run -all" tb_fibonacci_calculator  
 
The simulation transcript is also saved in a file named transcript.  

http://cwcserv.ucsd.edu/%7Ebilllin/classes/ECE111/examples/tb_fibonacci_calculator.sv
http://cwcserv.ucsd.edu/%7Ebilllin/classes/ECE111/examples/fibonacci_calculator.sv

Modelsim in Command Line Mode 
Make sure that the command line shell can find executables vlib, vlog, 
and vsim by setting the appropriate environment PATH variable. This should 
normally be set when you install ModelSim. For example, it might be located here:  
 
C:\intelFPGA_lite\17.1\modelsim_ase\win32aloem 

• This lecture provides more ways of using 
functions 

 

• Recall that “functions” should only be used to 
describe “combinational logic”, same as the way 
combinational logic is described in 
“always_comb” (combinational always 
statements) 

More on Functions 

module ex_add(input logic [1:0] a, b, c,  

             output logic [1:0] q); 

  function logic [1:0] add3(input logic [1:0] x, y, z); 

  logic [1:0] t; 

  begin 

    t = x + y; 

    add3 = t + z; 

  end 

  endfunction 

 

  assign q = add3(a, b, c); 

endmodule 

Default Scope of Local Variables 

Be Careful !  “t” is shared across all invocation of add3 !!! 

module ex_add(input logic [1:0] a, b, c,  

             output logic [1:0] q1, q2); 

  function automatic logic [1:0] add3(input logic [1:0] 

    x, y, z); 

  logic [1:0] t; 

  begin 

    t = x + y; 

    add3 = t + z; 

  end 

  endfunction 

 

  assign q1 = add3(a, b, c); 

  assign q2 = add3(b, b, b); 

endmodule 

Automatic Functions 

“automatic” ensures that all local variables are truly local. 
Each invocation of “add3” will use a different “t”. 

Accessing Module Variables 

 

 

 

module ex_add(input logic [1:0] a, b, c,  

             output logic [1:0] q); 

  function logic [1:0] add3(input logic [1:0] x); 

  logic [1:0] t; 

  begin 

    t = a + x; 

    add3 = t + c; 

  end 

  endfunction 

 

  assign q = add3 (c); // will return a + c + c 

endmodule 

Functions can access module (global) variables 

module ex_add(input logic [1:0] a, b, c,  

             output logic [1:0] q); 

  function logic [1:0] add3(input logic [1:0] x, y, z); 

  logic [1:0] t; 

  begin 

    t = x + y; 

    return t + z; // instead of add3 = t + z; 

  end 

  endfunction 

 

  assign q = add3(a, b, c); 

endmodule 

Return on Functions 

module ex_add(input logic [1:0] a, b, c,  

             output logic [1:0] q); 

  function void add3(input logic [1:0] x, y, z, 

                    output logic [1:0] sum); 

  local [1:0] t; 

  begin 

    t = x + y; 

    sum = t + z; // instead of add3 = t + z; 

  end 

  endfunction 

 

  always_comb 

  begin 

    add3(a, b, c, q); 

  end 

endmodule 

Void Functions 

• The rest of this lecture will be a presentation 
entitled “Synthesizing SystemVerilog: Busting the 
Myth that SystemVerilog is only for Verification,” 
by  Stuart Sutherland and Don Mills 
 

• The presentation was given at the Synopsys User 
Group (SNUG) meeting in Silicon Valley  in 2013 
 

• Link to the companion paper is provided on the 
course schedule page 

More SystemVerilog Features 

What This Paper is About…

 Debunking a myth regarding SystemVerilog
 What constructs in SystemVerilog are synthesizable
 Why those constructs are important for you to use
 How well Design Compiler and Synplify-Pro support 

SystemVerilog synthesis
 Fifteen coding recommendations for getting the most from 

Synthesizable SystemVerilog
Only a few Synthesizable SystemVerilog constructs are discussed in this presentation; 

Refer to the paper for the full list and details of Synthesizable SystemVerilog

It’s a Myth!

Not True! – SystemVerilog was designed to enhance both the 
design and verification capabilities of traditional Verilog
 Technically, there is no such thing as “Verilog” – the IEEE 

changed the name to “SystemVerilog” in 2009
 VCS, Design Compiler and Synplify-Pro all support RTL 

modeling with SystemVerilog

Verilog is a design language, and 
SystemVerilog is a verification language

And synthesis 
compilers can’t 

read in 
SystemVerilog

Much of SystemVerilog is 
Intended to be Synthesizable

initial
disable
events 
wait # @
fork–join

$finish  $fopen  $fclose
$display  $write 
$monitor
`define  `ifdef  `else
`include  `timescale

wire  reg
integer  real
time
packed arrays
2D memory

+   =   *   / 
%
>>   <<

modules
parameters
function/tasks
always  @
assign

begin–end
while
for  forever
if–else
repeat

Verilog-1995 (created in 1984)

ANSI C style ports
generate
localparam
constant functions 

standard file I/O
$value$plusargs
`ifndef   `elsif    `line
@*

(* attributes *)
configurations
memory part selects
variable part select

multi dimensional arrays 
signed types
automatic
** (power operator)

Verilog-2001

SystemVerilog-2005/2009/2012

enum
typedef
structures
unions
2-state types
packages
$unit

++  -- +=  -=  *=  /= 
>>=  <<=  >>>=  <<<=
&=  |=  ^=  %=
==?  !=?
inside
streaming
casting

break 
continue
return 
do–while
case inside
aliasing
const

interfaces
nested hierarchy
unrestricted ports
automatic port connect
enhanced literals
time values and units
specialized procedures

packed arrays
array assignments
unique/priority case/if
void functions
function input defaults
function array args
parameterized types 

de
si

gn

assertions
test program blocks
clocking domains
process control

mailboxes
semaphores
constrained random values
direct C function calls

classes
inheritance
strings
references

dynamic arrays
associative arrays
queues
checkers

2-state types
shortreal type
globals
let macros

ve
rif

ic
at

io
n

uwire `begin_keywords `pragma $clog2
Verilog-2005

Part One:
SystemVerilog Declaration Enhancements

The Goal…
 Model more functionality in fewer lines of code
 Reduce redundancy
 Reduce the risk of coding errors

New Synthesizable
Variable Data Types

 Useful synthesizable variable types 
 logic — 4-state variable, user-defined size (replaces reg)
 enum — a variable with a specified set of legal values
 int — 32-bit 2-state var (use with for-loops, replaces integer)

Avoid 2-state types in 
synthesizable models – they can 

hide serious design bugs!

Although synthesizable, these types 
are best used in testbenches

 Other synthesizable variable types … not very useful in RTL
 bit — single bit 2-state variable
 byte — 8-bit 2-state variable
 shortint — 16-bit 2-state variable
 longint — 64-bit 2-state variable

 What’s the advantage?
 logic makes code more self-documenting (reg does 

not infer a “register,” but it looks like it does)
 The enum type is important – more on another slide

Simplified Port Type Rules

 Traditional Verilog has strict and confusing rules for port types
 Input ports must be a net type (wire) 
 Output ports must be:
 reg (a variable) if assigned from 

a procedural block (initial, always)
 wire if assigned from a continuous assignment 
 wire if driven by an instance of a module or primitive output

“logic” indicates the value set (4-state) to be simulated –
SystemVerilog infers a variable or net based on context

module chip
(input logic in1,
input logic in2,

output logic out1,
output logic out2
);

 SystemVerilog makes it easy…
 Just declare everything as logic !!!

module chip
(input wire in1,
input wire in2,
output reg out1,
output wire out2
);

 What’s the advantage?
 Creating and modifying modules just got a whole lot easier!

Enumerated Types

 SystemVerilog adds enumerated types to Verilog
 enum defines variables or nets with a legal set of values
 Each legal value is represented by a label
enum logic [2:0] {WAIT=3’b001, LOAD=3’b010, READY=3’b100} state;

 What’s the advantage?
 Enumerated types can prevent inadvertent (and hard to debug) 

coding errors (example on next slide)

 Enumerated types have strict rules
 The label value must be the same size as the variable
 Can be assigned a label from the enumerated list
 Can be assigned the value of an identical enumerated variable
 All other assignments are illegal

The Advantage of
Enumerated Variables

parameter [2:0]
WAIT = 3'b001,
LOAD = 3'b010,
DONE = 3'b001;

parameter [1:0]
READY = 3'b101,
SET   = 3'b010,
GO    = 3'b110;

reg [2:0] state, next_state; 
reg [2:0] mode_control; 

always @(posedge clk or negedge rstN)
if (!resetN) state <= 0;
else         state <= next_state;

always @(state)  // next state decoder
case (state)
WAIT : next_state = state + 1;
LOAD : next_state = state + 1;
DONE : next_state = state + 1;

endcase

always @(state) // output decoder
case (state)
WAIT : mode_control = READY;
LOAD : mode_control = SET;
DONE : mode_control = DONE;

endcase

Traditional Verilog

legal, but a bug – parameter size is too small

legal, but a bug – state+1 results in invalid 
state value

legal, but a bug – wrong reset value for state

legal, but a bug – wrong constant used for 
mode_control

legal, but a bug – WAIT and DONE have the 
same value

enum logic [2:0] 
{WAIT = 3'b001,
LOAD = 3'b010,
DONE = 3'b001}
state, next_state;

enum logic [1:0]
{READY = 3'b101,
SET   = 3'b010,
GO    = 3'b110}
mode_control;

always_ff @(posedge clk or negedge rstN)
if (!resetN) state <= 0;
else         state <= next_state;

always_comb // next state decoder
case (state)
WAIT : next_state = state + 1;
LOAD : next_state = state + 1;
DONE : next_state = state + 1;

endcase

always_comb // output decoder
case (state)
WAIT : mode_control = READY;
LOAD : mode_control = SET;
DONE : mode_control = DONE;

endcase

SystemVerilog

Structures

 SystemVerilog structures bundle multiple variables together
 The entire structure can be assigned a list of values
 Entire structure can copied to another structure of same type
 Entire structures can be passed through module ports

struct {
logic [ 7:0] opcode;
logic [31:0] data;
logic        status;

} operation; operation.data = 32’hFEEDFACE;

operation = ’{8’h55, 1024, 1’b0};

 What’s the advantage?
 Bundle related signals together under one name
 Reduce lines of RTL code substantially
 Reduce risk of declaration mismatches
 Can eliminate design errors often not found until late in a design 

cycle (inter-module mismatches, missed assignments, ...) 

Assign entire structure

Assign to structure member

User-defined Types

 SystemVerilog adds user-defined types to Verilog
 typedef defines a new type
 Can be based on built-in types or other user-defined types
 Variables and nets can be declared as a user-defined type

typedef logic [31:0] bus32_t;

typedef enum [7:0] {ADD, SUB, MULT, DIV, SHIFT, ROT, XOR, NOP} opcodes_t;   

typedef enum logic {FALSE, TRUE} boolean_t;

typedef struct {
opcodes_t opcode;
bus32_t data;
boolean_t status;

} operation_t;

module ALU (input  operation_t operation,
output bus32_t result);

operation_t registered_op;
...

endmodule

 What’s the advantage?
 Can define complex types 

once and use many times
 Ensures consistency 

throughout a module

Packages

 SystemVerilog adds a package construct to Verilog
 Allows the same definition to be used by many modules

 What’s the advantage?
 Ensures consistency throughout a project (including verification)
 Reduces duplicate code
 Makes code easier to maintain and reuse than `include
 Controlled scope

package project_types;

typedef logic [31:0] bus32_t;

typedef enum [7:0] {...} opcodes_t;   

typedef struct {...} operation_t;

function automatic crc_gen ...;

endpackage

module ALU

import project_types::*;

(input  operation_t operation,
output bus32_t result);

operation_t registered_op;
...

endmodule

Data Arrays

 Packed array (aka “vector”) enhancements
 Vectors can now be divided into sub fields

 Unpacked array enhancements
 Can now have arrays of structures, user-defined types, etc.
 C-like array declarations
 Assign to entire array at once
 Copy arrays
 Pass arrays through ports

logic [3:0][7:0] b; a 32-bit vector with  4  8-bit subfieds
b[3] b[2] b[1] b[0]

[7:0] [7:0] [7:0] [7:0]

 What’s the advantage?
 This is major! – Manipulating entire data arrays substantially 

reduces lines of code (see example on next page)

logic [7:0] a1 [0:1][0:3];

logic [7:0] a2 [2][4];

a1 = ’{’{7,3,0,5},’{default:’1}};

a2 = a1; copy entire array

assign values to entire array

C-like declaration

Working with Entire Arrays 
Reduces Lines of Code

module transmit_reg (output design_types::uni_t data_reg,
input  design_types::uni_t data_packet,
input  logic                clock, resetN);

always @(posedge clock or negedge resetN) 
if (!resetN) data_reg <= ’{default:0};
else         data_reg <= data_packet;

endmodule

package design_types;
typedef struct {
logic [ 3:0] GFC;
logic [ 7:0] VPI;
logic [15:0] VCI;
logic        CLP;
logic [ 2:0] T;
logic [ 7:0] HEC;
logic [ 7:0] Payload [48];

} uni_t; // UNI cell definition
endpackage 54 ports in old Verilog

another 54 ports

54 separate assignment 
statements in old Verilog

This structure bundles 54 
variables together (including the 

array of 48 Payload variables)

another 54 assignments

 What’s the advantage?
 4 lines of code in SystemVerilog replaces 216 lines of 

old Verilog – and ensures consistency in all 4 places!

54 more separate assignment 
statements in old Verilog

Interface Ports

 SystemVerilog interfaces are a compound, multi-signal port
 Bundles any number of signals (nets and variables) together
 Bundles “methods” (tasks and functions) with the signals
 Bundles assertion checks with the signals

interface chip_bus;
logic [31:0] data, address;
logic        request, grant,
boolean_t ready;

endinterface

module CPU (chip_bus bus,
input logic clk,
input logic reset);

...

 What’s the advantage?
 Simplifies complex bus definitions and interconnections
 Ensures consistency throughout the design

RAM
clk clk

data data
address address
request request

grant grant
ready ready

CPU

Verilog discrete ports

reset reset

mclk mrst

SystemVerilog interface ports

interface
port

interface
port

chip_bus
interface

RAM
clk clk

CPU

reset reset

mclk mrst

Part Two:
SystemVerilog Programming Enhancements

The Goal…
 Model RTL functionality more accurately
 Reduce mismatches in RTL simulation vs. synthesized gates
 Fewer lines of code – concisely model complex functionality

Hardware Specific 
Procedural Blocks

 SystemVerilog adds special hardware-oriented procedures: 
always_ff, always_comb, and always_latch
 Document engineer’s intent
 Software tool can verify that functionality meets the intent
 Enforce several semantic rules required by synthesis
always @(mode)
if (!mode)
o1 = a + b;

else
o2 = a - b;

Traditional Verilog
Synthesis must 

guess (infer) what 
type of logic was 

intended

always_comb
if (!mode)
o1 = a + b;

else
o2 = a - b;

SystemVerilog
Contents checked 
for adherence to 

synthesis rules for 
combinational logic

 What’s the advantage?
 RTL code intent is self-documented
 Non-synthesizable code won’t simulate
 Simulation, synthesis and formal tools use same rules

The case( ) inside
Decision Statement

 The case() inside statement replaces casex and casez
 Bits set to X, Z or ? in the case items are “don’t care” bits 
 Any X, Z or ? bits in the case expression are not don’t cares 
 With casez and casex, X, Z of ? bits in the case expression are 

also considered don’t cares – which is a serious problem
case (opcode) inside 
8’b1???????: ... // only compare most significant bit
8’b????1111: ... // compare lower 4 bits, ignore upper bits
...
default: $error("bad opcode");

endcase
If opcode has the value 8'bzzzzzzzz, which branch should execute?

 What’s the advantage?
 case() inside eliminates the serious GOTCHA of casex

and casez than could lead to design bugs going undetected

Unique and Priority Decisions

 The unique, unique0 and priority decision modifiers…
 Enable parallel_case and/or full_case synthesis pragmas
 Enable run-time simulation checking for when the decision might 

not work as expected if synthesized with the pragma
• Enables full_case and parallel_case pragmas
• Will get simulation warnings if state matches 

multiple branches (not a valid parallel_case)
• Will get simulation warnings if state doesn’t 

match any branch (not a valid full_case)

always_comb
unique case (state)

RDY: ...
SET: ...
GO : ... 

endcase

 What’s the advantage?
 Automatic run-time checking that the decision statement will 

synthesize as intended

WARNING: These decision modifiers do not eliminate the evil side of the full_case
and parellel_case twins –– but, the keywords do warn about the presence of evil

Operators

 SystemVerilog adds many new synthesizable constructs:
 ==? and !=? wildcard equality/inequality operators
 inside set membership operator 
 <<, >> pack and unpack streaming operators
 ++ and -- increment and decrement operators
 +=, -=, *=, /= …  assignment operators

 What’s the advantage?
 Model more RTL functionality in fewer lines of code

a = { << { b }};

if data is between 0 to 255, inclusive

if data is 3'b101, 3'b111, 3'b1x1, or 3'b1z1

bit reverse – unpack bits of b and assign to a in reverse order

if (data inside {[0:255}) ...

if (data inside {3'b1?1}) ...

c = { <<8{ d }}; byte reverse – unpack 8-bit chunks of d and assign in reverse order

How much Verilog code would 
these operations require?

Type, Size and Sign Casting

 SystemVerilog adds casting operations to Verilog
 <type>’(<expression>) — cast expression to different data type
 <size>’(<expression>) — casts expression to a vector size
 signed’(<expression>) — casts expression to signed
 unsigned’(<expression>) — casts expression to unsigned

 What’s the advantage?
 Documents intent that a change in type, size or sign is intended
 Can eliminate size and type mismatch warnings

cast the operation result to 32 bits so that 
the RHS and the LHS are the same sizey = logic [31:0]'({a,a} >> b);

Rotate a by b 
number of times

logic [31:0] a, y;
logic [ 5:0] b;

y = {a,a} >> b;

Will get warning from lint checkers 
and synthesis because LHS is 32 

bits and RHS is 64 bits

Module Instance
Port Connection Shortcuts

 Verilog netlist port connections must name both the port and 
the net connected to it module dff (output q, qb,

input  clk, d, rst, pre);
...

endmodulemodule chip (output [3:0] q,
input  [3:0] d, input clk, rst, pre);

dff dff1 (.clk(clk), .rst(rst), .pre(pre), .d(d[0]), .q(q[0]));

can be verbose and redundant

dff dff1 (.clk, .rst, .pre, .d(d[0]), .q(q[0]));

 SystemVerilog adds .name and .* shortcuts
 .name connects a port to a net of the same name

dff dff1 (.*, .q(q[0]), .d(d[0]), .qb());

 .* automatically connects all ports and nets with the same name

 What’s the advantage?
 Reduce typing (and typos) when connecting design blocks
 Built-in checking prevents connection mismatches

Enhanced Literal Value 
Assignments

 In Verilog, there is no simple way to fill a vector with all 1’s
parameter N = 64;
reg [N-1:0] data_bus;
data_bus = 64’hFFFFFFFFFFFFFFF; //set all bits of data_bus to 1

vector width must be hard coded

could also use coding tricks, such 
as replicate or invert operations

reg [N-1:0] data_bus;
data_bus = x’1;

 SystemVerilog adds a vector fill literal value
x’0 fills all bits on the left-hand side with 0
x’1 fills all bits on the left-hand side with 1
x’z fills all bits on the left-hand side with z
x’x fills all bits on the left-hand side with x

set all bits of data_bus to 1

 What’s the advantage?
 Code will scale correctly when vector sizes change
 Don’t need to know obscure coding tricks such as replicate

Verilog and SystemVerilog 
Compatibility Directives

 SystemVerilog is backward compatible with Verilog
 Old Verilog and SystemVerilog models can be intermixed
 SystemVerilog does add many keywords to Verilog
 In Verilog models, those keywords were legal to use as names
 The `begin_keywords directive tells software tools which 

version of reserved keywords to use during compilation
`begin_keywords 1364-2001
module test;
wire priority; 
...

endmodule
`end_keywords

`begin_keywords 1800-2005
module decoder (...);
always_comb
priority case (...); 
...

endmodule
`end_keywords

In Verilog “priority” is 
not a reserved keyword

In SystemVerilog “priority”
is a reserved keyword

 What’s the advantage?
 Ensures design code is reusable, past, present and future

Lots of Enhancements to
Tasks and Functions

 SystemVerilog enhancements tasks and functions several ways
 Void functions – this one is important for synthesis!
 Functions with output and inout formal arguments
 Formal arguments default to input
 Arrays, structures, user-defined types as formal arguments
 Pass by name in task/function calls
 Function return values can be specified, using return
 Parameterized task/function arguments using static classes

 What’s the advantage?
 Fewer lines of code
 Reusable code

Recommendation – use void 
functions instead of tasks in 

synthesizable models

Part Three:
Synthesis Considerations

The paper also discusses…
 Design Compiler versus Synplicity-Pro
 Some things that should be synthesizable
 15 recommendations for how you can benefit from 

SystemVerilog

Differences between Design 
Compiler and Synplicity-Pro

 DC and Synplify-Pro are closely aligned, but there are some 
differences in the SystemVerilog constructs supported

SystemVerilog Construct Design Compiler 
2012.06-SP4

Synplify-Pro 
2012.09

‘begin_keyword, ‘end_keyword compatibility directives yes no

Package import before module port list yes no

case...inside yes no

priority, unique0 and unique modifier to if...else yes ignored

Parameterized tasks and functions (using classes) yes no

real data type no yes

Nets declared from typedef struct definitions no yes

Immediate assertions ignored yes

Interface modport expressions no yes

Several important differences are listed in this table – refer 
to the paper for a more complete list of differences



28 of 30
Stu Sutherland
Sutherland HDL

Don Mills
Microchip

DC and/or Synplicity-Pro
Wish List

 SystemVerilog has several constructs that are useful for 
modeling hardware, but which are not synthesizable
 uwire single source nets
 foreach loops
 Task/function inputs with default values
 Task/function ref arguments
 Set membership operator (inside) with expressions
 Package chaining
 Extern module declarations
 Configurations
 Generic and user-defined net types

Let your Synopsys rep know if 
any of these features would 
help you in your projects!

Fifteen Ways You Can Benefit from 
Using SystemVerilog in RTL designs

1. Use logic for modules ports and most internal signals – forget wire, reg
2. Use the uwire net type to check for and enforce single-driver logic
3. Use enumerated types for variables with limited legal values
4. Use structures to collect related variables together
5. Use user-defined types to ensure consistent declarations in a design
6. Use packages for declarations that are shared throughout a design
7. Use always_comb, always_latch and always_ff procedural blocks
8. Use case...inside instead of casez and casex
9. Use priority, unique0, unique instead of full_case, parallel_case
10. Use priority, unique0, unique with if...else when appropriate
11. Use void function instead of task in RTL code
12. Use dot-name and dot-star netlist shortcuts
13. Use interfaces to group related bus signals
14. Use `begin_keywords to specify the language version used
15. Use a locally declared timeunit instead of `timescale

Summary

 It’s a myth – SystemVerilog is not just for verification, it is also a 
synthesizable design language
 Technically, there is no such thing as “Verilog” – the IEEE 

changed the name to “SystemVerilog” in 2009
 SystemVerilog adds many important synthesizable constructs 

to the old Verilog language
 Design more functionality in fewer lines of code
 Ensure RTL code will synthesize to the logic intended
 Make code more reusable in future projects
 Design Compiler and Synplify-Pro both support SystemVerilog
 There are some differences (see the paper for details)
 There are many benefits to using SystemVerilog for ASIC and 

FPGA design



Questions?Questions?

the answer is in the paper ... somewhere
( if not, we’ll find out  )

Stu Sutherland
stuart@sutherland-hdl.com

Don Mills
mills@microchip.com
mills@lcdm-eng.com

